version 15.1

cd "G:\Experiment Data"
use Experiment2_encoded_data.dta

**Regression of Performance Ratings on IAT Scores (Table 1)
reg PerformanceRating IATscore

**The Interaction of Information and Implicit Attitudes in the Evaluative Process
**Centralize IAT Scores
center IATscore
**Use the newly generated variable c_IATscore for regression analysis (Table 2)
reg PerformanceRating c.c_IATscore##i.Treat

**The Interactive Effect of Information and Time on Performance Ratings (Table 3)
reg PerformanceRating i.Treat

**Test of the difference between the effect of information and advertising on performance ratings (Table 4)
test 1.Treat = 2.Treat

**Descriptive Statistics(please interpret the results in conjunction with the codebook)
**For continuous variable (Table S2)
sum IATscore Age

**For categorical variables (Table S2)
tab Gender
tab Region
tab UrbanRural
tab MaritalStatus
tab FertilityStatus
tab Nation
tab Education
tab PoliticalStatus
tab Income

**Balance test across experimental groups (Table S4)
oneway Gender Treat, tabulate
oneway Region Treat, tabulate
oneway UrbanRural Treat, tabulate
oneway Age Treat, tabulate
oneway MaritalStatus Treat, tabulate
oneway FertilityStatus Treat, tabulate
oneway Nation Treat, tabulate
oneway Education Treat, tabulate
oneway PoliticalStatus Treat, tabulate
oneway Income Treat, tabulate
oneway Prior_EMS_use Treat, tabulate
oneway Prior_ZTOEx_use Treat, tabulate
oneway GovernmentTrust Treat, tabulate

**The Interaction of Information and Implicit Attitudes in the Evaluative Process: //Implicit Attitudes as a Moderator (Table S7)
**Recoding IATscore: 
//IAT_indicator = 0 if IATscore < 0; 
//IAT_indicator = 1 if IATscore > 0. 
//IAT scores that are greater than zero indicate unfavorable implicit public sector attitudes. 
//IAT scores that are less than zero indicate favorable implicit public sector attitudes.
gen IAT_indicator1 = round(IATscore, 0.01)
replace IAT_indicator1 = . if IAT_indicator1 == 0
gen IAT_indicator = cond(IAT_indicator1 < 0,0,1)
replace IAT_indicator = . if IAT_indicator1 == .
drop IAT_indicator1

reg PerformanceRating i.Treat##i.IAT_indicator

**Potential moderators of information effects (Table S8 and Table S10)
// Recoding and Preparing Variables

**Recoding Treat:
//1 if a subject received information or advertising
//0 otherwise
recode Treat (0=0) (1/2=1), generate(Information)

**Recoding Education: 
//1 if 4-year degree or more
//0 if less than 4-year degree
recode Education (0=0) (1/2=1), generate(New_Education)

**Recoding Income: 
//1 if more than ¥100,000 
//0 if less than or equal to ¥100,000
recode Income (0/1=0) (2/3=1), generate(New_Income)

**Recoding GovernmentTrust: 
//1 if trust just about always or most of the time
//0 if trust only some of the time or almost never
recode GovernmentTrust (3/4=0) (1/2=1), generate(Trust)

**Recoding Prior_EMS_use:
//1 if frequent EMS use 
//0 if infrequent use
recode Prior_EMS_use (1/3=0) (4/5=1), generate(NewPriorEMS_use)

**Recoding PoliticalStatus:
//1 if Communist 
//0 if non-Communist
recode PoliticalStatus (0/1=0) (2=1) (3=0), generate(New_PoliticalStatus)

**The Interactive Effect of Information and Potential moderators on Performance Ratings
reg PerformanceRating i.Information##i.New_Education
reg PerformanceRating i.Information##i.New_Income
reg PerformanceRating i.Information##i.Trust
reg PerformanceRating i.Information##i.NewPriorEMS_use
reg PerformanceRating i.Information##i.New_PoliticalStatus
reg PerformanceRating i.Information##i.Region
